Задача 8. Дана прямоугольная таблица из нулей и единиц – просканированный рисунок отпечатка древнего листа на угле. Найти количество сторон многоугольника. 

Пример входного файла: 

10101
01110

00100

01110

11111

Пример выходного файла: 

9
Решение. Будем подсчитывать количество сторон многоугольника, буквально проходя по его сторонам, например, против часовой стрелки. Заметим, что число вершин многоугольника совпадает с числом его сторон. На самом деле, пудем считать количество вершин многоугольника. 
Сначала найдём левую верхнюю вершину многоугольника. Для этого просматриваем строки слева направо, первую строку, втору строку и т. д. до тех пор, пока нам не встретится первая единица. Запоминаем её координаты. Это и есть левая верхняя вершина многоугольника. 

Пронумеруем все возможные направления, например так:

вверх – 1;

вверх влево – 2;

влево – 3;

вниз влево – 4;

вниз – 5;

вниз вправо – 6;

вправо – 7;

вверх вправо – 8.

Всего – 8 направлений. 

Теперь начинаем движение от первой точки. Начинаем «оглядываться», перебирая направления начиная от направления 2 (для первой точки там заведомо нет единицы), постепенно увеличивая номер направления и проверяем, есть ли единица в этом направлении. При увеличении направления следим, чтобы его номер был не больше 8. Если при увеличении направления номер направления равен 9, то номер направления следует установить на 1. Таким образом, мы «оглядываемся» против часовой стрелки, пока не найдём направление дальнейшего движения. Если новое направление не равно предыдущему направлению, то счётчик вершин увеличиваем на единицу. Теперь двигаемся в новом направлении. Для этого напишем процедуру движения от данной точки в данном направлении. Если же новое направление и старое направление совпадают,  то просто двигаемся в прежнем направлении.
Будем повторять описанную выше последовательность действий до тех пор, пока не встанем на исходную точку, координаты которой мы предусмотрительно сохранили. Остаётся теперь только вывести готовый ответ: число подсчитанных сторон многоугольника.
Описанный выше алгоритм позволяет не только подсчитать количество сторон многоугольника, но даже нарисовать его! Это действительно теперь не сложно: для этого в процедуру движения в данном направлении включим команду рисования отрезка фиксированной длинны в том же направлении.
Теперь программа умет ещё и рисовать данный лист!

Ниже приведён исходный код программы целиком.

program qq;

  uses crt,graph; {Библиотека Graph для рисования данного листа :) }

  const nn=30; {Максимальный размер сканируемого прямоугольника}

  Type List=array[1..nn,1..nn] of integer; {Тип для хранения сканированного рисунка}

  var f,g:Text; {Переменные для входных и выходных данных}

      i,j, {Переменные индексы. Также используются для хранения

            текущих координат на стороне литса при обходе листа

            против часовой стрелки}

      i0,j0, {координаты начальной точки. Служат для проверки того факта,

              что при обходе листта вернулись в начальную точку.

              Эта начальная точка - правый верхний угол листа}

      s,code, {переменные для преобразования строк в число}

      v, {Переменная - счётчик для подсчёта сторон многоугольника

         Будем аккуратно увеличивать на 1 при обнаружении новой стороны}

      dir1,dir2, {переменные для хранения направлений, предыдущее и последующее:

                  1 - вверх; 2 - вверх и влево и т.д. против часовой стрелке;

                  8 - вверх и вправо

                  При смене направлений увеличиваем счётчик сторон v на 1}

      gd,gm,{переменные для включения графического режима}

      p,q,{Текущие координаты на экране при рисовании листа в графическом

           режиме}

      n,{Высота прямоугольника. Будет определяться автоматически в

         зависимости от количества прочитанных строк}

      mm:integer; {Длина прямоугольника. Будет определяться автоматически

                   как длина самой длинной строки}

      h:string; {переменная для чтения строк из входного файла}

      x:List; {Матрица из нулей и единиц}

      L:Boolean; {Флаг, признак того, что не найден правый верхний угол листа}

  Function Check(i,j,dir:integer):Boolean;

    {Проверяет, есть ли 1 в данном направлении}

    var L:Boolean;

    begin

      case dir of

        1:if i=1 then L:=False else L:=(x[i-1,j]=1); {Вверх}

        2:if (i=1)or(j=1) then L:=False else L:=(x[i-1,j-1]=1); {Вверх Лево}

        3:if j=1 then L:=False else L:=(x[i,j-1]=1); {Влево}

        4:if (i=n)or(j=1) then L:=False else L:=(x[i+1,j-1]=1); {Вниз Влево}

        5:if (i=n) then L:=False else L:=(x[i+1,j]=1); {Вниз}

        6:if (i=n)or(j=mm) then L:=False else L:=(x[i+1,j+1]=1); {Вниз Вправо}

        7:if j=mm then L:=False else L:=(x[i,j+1]=1); {Вправо}

        8:if (i=1)or(j=mm) then L:=False else L:=(x[i-1,j+1]=1); {Вверх Вправо}

      end;

      Check:=L

    end;

  procedure Step(var i,j:integer; dir:integer);

    {Делает 1 шаг из точки с координатами (i,j) в данном направлении dir}

    begin

      Case dir of

        1:dec(i); {Шаг вверх}

        2:begin dec(i); dec(j) end; {Шаг вверх влево}

        3:dec(j); {Шаг влево}

        4:begin inc(i); dec(j) end; {Шаг вниз влево}

        5:inc(i);{Шаг вниз}

        6:begin inc(i); inc(j) end;{Шаг вниз вправо}

        7:inc(j);{Шаг вправо}

        8:begin dec(i); inc(j) end;{Шаг вверх вправо}

      end

    end;

  procedure stepgraph(var p,q:integer;dir:integer);

    {Рисует лист, вернее, строит очередную сторону многоугольника-листа}

    const m=25; {Длина одного шага}

    begin

      Case dir of

        1:begin p:=p-m; lineto (q,p) end; {Шаг вверх}

        2:begin p:=p-m; q:=q-m;lineto (q,p)end; {Шаг вверх влево}

        3:begin q:=q-m;lineto (q,p) end; {Шаг влево}

        4:begin p:=p+m; q:=q-m;lineto (q,p) end; {Шаг вниз влево}

        5:begin p:=p+m;lineto (q,p) end; {Шаг вниз}

        6:begin p:=p+m; q:=q+m;lineto (q,p) end;{Шаг вниз вправо}

        7:begin q:=q+m;lineto (q,p)end;{Шаг вправо}

        8:begin p:=p-m; q:=q+m;lineto (q,p) end;{Шаг вверх вправо}

      end

    end;

  Function card(x:List):integer; {Считает количество единиц в прямоугольнике

    и защищает от зацикливания, если таблице только 1 единица или только нули}

    var s,i1,j1:integer;

    begin

      s:=0; {Счётчик единиц}

      for i1:=1 to n do for j1:=1 to mm do if x[i1,j1]=1 then inc(s);

      card:=s

    end;

begin

  TextBackGround(1);  TextColor(14);  ClrScr;

  Assign(f,'c:/List.in');  reset(f);  n:=1; mm:=0; {Готовы читать строки данных}

  while not(eof(f))and (i<nn) do

    begin

      readln(f,h);

      if length(h)>mm then mm:=length(h);{Максимальная длина строки из 0 и 1}

      for j:=1 to length(h) do

        begin

          val(h[j],s,code); x[n,j]:=s

        end;

      inc(n)

    end;

  dec(n); close(f);{Данные прочитаны, матрица из 0 и 1 построена}

  for i:=1 to n do

    begin

      for j:=1 to mm do write(x[i,j]);

      writeln

    end;

  readln; TextColor(11); {Поиск верхнего правого угла}

  i:=1; j:=1;  L:=True;

  while L and (i<=n) do

    begin

      while (x[i,j]=0)and(j<mm) do inc(j);

      if x[i,j]=1

        then

          L:=False

        else

          begin

            inc(i);j:=1;

          end

    end;

  i0:=i;j0:=j; {Координаты верхнего правого угла}

  dir1:=3;

  v:=0; {Счетчик сторон}

  if card(x)>1

    then

      begin

        gd:=0;gd:=0;

        initgraph(gd,gm,'c:/tp/bgi');

        p:=100;q:=100; {Начало рисунка листа}

        Setcolor(11);

        moveto(q,p);

        repeat

          dir2:=dir1-2; if dir2<=0 then inc(dir2,8);

          while not(check(i,j,dir2)) do

            begin

              inc(dir2);

              if dir2>8 then dir2:=dir2-8;

            end;

          {Выбрано направление}

          if dir1<>dir2 then inc(v); {Посчитали еще одну сторону}

          {writeln(v);}

          {Направление выбрано, делаем шаг в направлении dir2}

          Step(i,j,dir2);

          StepGraph(p,q,dir2);

          dir1:=dir2; {Шаг завершен}

        until (i=i0)and(j=j0);

        readln; closegraph;

        ClrScr;

  end

  else

    begin

      writeln('Лист слишком мал, не более 1 точки')

    end;

  assign(g,'c:/List.Out'); {Вывод результата}

  rewrite(g); write(g,v);  write(v);  close(g);

  repeat until keypressed;

end.

